
1 Fixed points and contraction mappings
As discussed in previous lectures, many iterations we consider have the form

xk+1 = G(xk)

where G : Rn → Rn. We call G a contraction on Ω if it is Lipschitz with
constant less than one, i.e.

‖G(x)−G(y)‖ ≤ α‖x− y‖, α < 1.

A sufficient (but not necessary) condition for G to be Lipschitz on Ω is if G
is differentiable and ‖G′(x)‖ ≤ α for all x ∈ Ω.

According to the contraction mapping theorem or Banach fixed point the-
orem, when G is a contraction on a closed set Ω and G(Ω) ⊂ Ω, there is a
unique fixed point x∗ ∈ Ω (i.e. a point such that x∗ − G(x∗)). If we can ex-
press the solution of a nonlinear equation as the fixed point of a contraction
mapping, we get two immediate benefits.

First, we know that a solution exists and is unique (at least, it is unique
within Ω). This is a nontrivial advantage, as it is easy to write nonlinear
equations that have no solutions, or have continuous families of solutions,
without realizing that there is a problem.

Second, we have a numerical method – albeit a potentially slow one – for
computing the fixed point. We take the fixed point iteration

xk+1 = G(xk)

started from some x0 ∈ Ω, and we subtract the fixed point equation x∗ =
G(x∗) to get an iteration for ek = xk − x∗:

ek+1 = G(x∗ + ek)−G(x∗)

Using contractivity, we get
‖ek+1‖ = ‖G(x∗ + ek)−G(x∗)‖ ≤ α‖ek‖

which implies that ‖ek‖ ≤ αk‖e0‖ → 0.
When error goes down by a factor of α > 0 at each step, we say the itera-

tion is linearly convergent (or geometrically convergent). The name reflects a
semi-logarithmic plot of (log) error versus iteration count; if the errors lie on
a straight line, we have linear convergence. Contractive fixed point iterations
converge at least linarly, but may converge more quickly.

1.1 A toy example
Consider the function G : R2 → R2 given by

G(x) =
1

4

[
x1 − cos(x2)
x2 − sin(x1)

]
This is a contraction mapping on all of R2 (why?).

Let’s look at ‖xk − G(xk)‖ as a function of k, starting from the initial
guess x0 = 0 (Figure 1).

function test_toy_contraction()

G(x) = 0.25*[x[1]-cos(x[2]); x[2]-sin(x[1])]

# Run the fixed point iteration for 100 steps

resid_norms = []

x = zeros(2)

for k = 1:100

x = G(x)

push!(resid_norms, norm(x-G(x), Inf))

end

x, resid_norms

end

1.1.1 Questions

1. Show that for the example above ‖G′‖∞ ≤ 1
2

over all of R2. This
implies that ‖G(x)−G(y)‖∞ ≤ 1

2
‖x− y‖∞.

2. The mapping x 7→ x/2 is a contraction on (0,∞), but does not have a
fixed point on that interval. Why does this not contradict the contrac-
tion mapping theorem?

3. For S > 0, show that the mapping g(x) = 1
2
(x+S/x) is a contraction on

the interval [
√
S,∞). What is the fixed point? What is the Lipschitz

constant?

Figure 1: Convergence for test_toy_contraction.

2 Newton’s method for nonlinear equations
The idea behind Newton’s method is to approximate a nonlinear f ∈ C1

by linearizations around successive guesses. We then get the next guess by
finding where the linearized approximaton is zero. That is, we set

f(xk+1) ≈ f(xk) + f ′(xk)(xk+1 − xk) = 0,

which we can rearrange to

xk+1 = xk − f ′(xk)−1f(xk).

Of course, we do not actually want to form an inverse, and to set the stage
for later variations on the method, we also write the iteration as

f ′(xk)pk = −f(xk)

xk+1 = xk + pk.

2.1 A toy example
Consider the problem of finding the solutions to the system

x+ 2y = 2

x2 + 4y2 = 4.

Figure 2: Convergence for test_toy_newton to (0, 1).

That is, we are looking for the intersection of a straight line and an ellipse.
This is a simple enough problem that we can compute the solution in closed
form; there are intersections at (0, 1) and at (2, 0). Suppose we did not know
this, and instead wanted to solve the system by Newton’s iteration. To do
this, we need to write the problem as finding the zero of some function

f(x, y) =

[
x+ 2y − 2
x2 + 4y2 − 4

]
= 0.

We also need the Jacobian J = f ′:

∂f

∂(x, y)
=

[
∂f1
∂x

∂f1
∂y

∂f2
∂x

∂f2
∂y

]
.

We show the convergence of the Newton iteration in Figure 2.

function test_toy_newton(x, y)

# Set up the function and the Jacobian

f(x) = [x[1] + 2*x[2]-2;

x[1]^2 + 4*x[2]^2 - 4]

J(x) = [1 2 ;

2*x[1] 8*x[2]]

# Run ten steps of Newton from the initial guess

x = [x; y]

fx = f(x)

resids = zeros(10)

for k = 1:10

x -= J(x)\fx

fx = f(x)

resids[k] = norm(fx)

end

x, resids

end

2.1.1 Questions

1. Finding an (real) eigenvalue of A can be posed as a nonlinear equation
solving problem: we want to find x and λ such that Ax = λx and
xTx = 1. Write a Newton iteration for this problem.

2.2 Superlinear convergence
Suppose f(x∗) = 0. Taylor expansion about xk gives

0 = f(x∗) = f(xk) + f ′(xk)(x∗ − xk) + r(xk)

where the remainder term r(xk) is o(‖xk − x∗‖) = o(‖ek‖). Hence

xk+1 = x∗ + f ′(xk)−1r(xk)

and subtracting x∗ from both sides gives

ek+1 = f ′(xk)−1r(xk) = f ′(xk)−1o(‖ek‖)

If ‖f ′(x)−1‖ is bounded for x near x∗ and x0 is close enough, this is sufficient
to guarantee superlinear convergence. When we have a stronger condition,
such as f ′ Lipschitz, we get quadratic convergence, i.e. ek+1 = O(‖ek‖2). Of
course, this is all local theory – we need a good initial guess!

Figure 3: Two solutions for the blow-up example.

3 A more complex example
We now consider a more serious example problem, a nonlinear system that
comes from a discretized PDE reaction-diffusion model describing (for exam-
ple) the steady state heat distribution in a medium going an auto-catalytic
reaction. The physics is that heat is generated in the medium due to a reac-
tion, with more heat where the temperature is higher. The heat then diffuses
through the medium, and the outside edges of the domain are kept at the
ambient temperature. The PDE and boundary conditions are

d2v

dx2
+ exp(v) = 0, x ∈ (0, 1)

v(0) = v(1) = 0.

We discretize the PDE for computer solution by introducing a mesh xi = ih
for i = 0, . . . , N + 1 and h = 1/(N + 1); the solution is approximated by
v(xi) ≈ vi. We set v0 = vN+1 = 0; when we refer to v without subscripts,
we mean the vector of entries v1 through vN . This discretization leads to the
nonlinear system

fi(v) ≡
vi−1 − 2vi + vi+1

h2
+ exp(vi) = 0.

This system has two solutions; physically, these correspond to stable and
unstable steady-state solutions of the time-dependent version of the model.

We show the two solutions in Figure 3.

Figure 4: Newton convergence for different starting guesses.

To solve for a Newton step, we need both f and the Jacobian of f with
respect to the variables vj. This is a tridiagonal matrix, which we can write
as

J(v) = −h−2TN + diag(exp(v))
where TN ∈ RN×N is the tridiagonal matrix with 2 down the main diagonal
and −1 on the off-diagonals.

function autocatalytic(v)

N = length(v)

fv = exp.(v)

fv -= 2*(N+1)^2*v

fv[1:N-1] += (N+1)^2*v[2:N ]

fv[2:N ] += (N+1)^2*v[1:N-1]

fv

end

function Jautocatalytic(v)

N = length(v)

SymTridiagonal(exp.(v) .- 2*(N+1)^2, (N+1)^2 * ones(N-1))

end

For an initial guess, we use vi = αxi(1− xi) for different values of α. For
α = 0, we converge to the stable solution; for α = 20 and α = 40, we converge
to the unstable solution (Figure 3). We eventually see quadratic convergence
in all cases, but for α = 40 there is a longer period before convergence sets
in (Figure 4). For α = 60, the method does not converge at all.

Figure 5: Newton convergence vs fixed point convergence.

function newton_autocatalytic(α, N=100, nsteps=50, rtol=1e-8;

monitor=(v, resid)->nothing)

v_all = [α*x*(1-x) for x in range(0.0, 1.0, length=N+2)]

v = v_all[2:N+1]

for step = 1:nsteps

fv = autocatalytic(v)

resid = norm(fv)

monitor(v, resid)

if resid < rtol

v_all[2:N+1] = v

return v_all

end

v -= Jautocatalytic(v)\fv

end

error("Newton did not converge after $nsteps steps")

end

We can derive a Newton-like fixed point iteration from the observation
that if v remains modest, the Jacobian is pretty close to −h2TN . This gives
us the iteration

h−2TNv
k+1 = exp(vk).

In Figure 5, we compare the convergence of this fixed point iteration to
Newton’s method. The fixed point iteration does converge, but it shows the
usual linear convergence, while Newton’s method converges quadratically.

function fp_autocatalytic(α, N=100, nsteps=500, rtol=1e-8;

monitor=(v, resid)->nothing)

v_all = [α*x*(1-x) for x in range(0.0, 1.0, length=N+2)]

v = v_all[2:N+1]

TN = SymTridiagonal(2.0*ones(N), -ones(N-1))

F = ldlt(TN)

for step = 1:nsteps

fv = autocatalytic(v)

resid = norm(fv)

monitor(v, resid)

if resid < rtol

v_all[2:N+1] = v

return v_all

end

v[:] = F\(exp.(v)/(N+1)^2)

end

error("Fixed point iteration did not converge after $nsteps steps (α=$α)")

end

3.0.1 Questions

1. Consider choosing v = αx(1−x) so that the equation is exactly satisfied
at x = 1/2. How would you do this numerically?

2. Modify the Newton solver for the discretization of the equation v′′ +
λ exp(v) = 0. What happens as λ grows greater than one? For what
size λ can you get a solution? Try incrementally increasing λ, using
the final solution for the previous value of λ as the initial guess at the
next value.

4 Some practical issues
In general, there is no guarantee that a given solution of nonlinear equations
will have a solution; and if there is a solution, there is no guarantee of unique-
ness. This has a practical implication: many incautious computationalists
have been embarrassed to find that they have “solved” a problem that was
later proved to have no solution!

When we have reason to believe that a given system of equations has a
solution — whether through mathematical argument or physical intuition —
we still have the issue of finding a good enough initial estimate that Newton’s
method will converge. In coming lectures, we will discuss “globalization”
methods that expand the set of initial guesses for which Newton’s method
converges; but globalization does not free us from the responsibility of trying
for a good guess. Finding a good guess helps ensure that our methods will
converge quickly, and to the “correct” solution (particularly when there are
multiple possible solutions).

We saw one explicit example of the role of the initial guess in our analysis
of the discretized blowup PDE problem. Another example occurs when we
use unguarded Newton iteration for optimization. Given a poor choice of
initial guess, we are as likely to converge to a saddle point or a local maximum
as to a minimum! But we will address this pathology in our discussion of
globalization methods.

If we have a nice problem and an adequate initial guess, Newton’s itera-
tion can converge quite quickly. But even then, we still have to think about
when we will be satisfied with an approximate solution. A robust solver
should check a few possible termination criteria:

• Iteration count: It makes sense to terminate (possibly with a diagnostic
message) whenever an iteration starts to take more steps than one
expects — or perhaps more steps than one can afford. If nothing else,
this is necessary to deal with issues like poor initial guesses.

• Residual check: We often declare completion when ‖f(xk)‖ is suffi-
ciently close to zero. What is “close to zero” depends on the scaling
of the problem, so users of black box solvers are well advised to check
that any default residual checks make sense for their problems.

• Update check: Once Newton starts converging, a good estimate for the
error at step xk is xk+1 − xk. A natural test is then to make sure
that ‖xk+1 − xk‖/‖xk+1‖ < τ for some tolerance τ . Of course, this is
really an estimate of the relative error at step k, but we will report the
(presumably better) answer xk+1 – like anyone else who can manage it,
numerical analysts like to have their cake and eat it, too.

A common problem with many solvers is to make the termination criteria
too lax, so that a bad solution is accepted; or too conservative, so that good
solutions are never accepted.

One common mistake in coding Newton’s method is to goof in computing
the Jacobian matrix. This error is not only very common, but also very
often overlooked. After all, a good approximation to the Jacobian often still
produces a convergent iteration; and when iterations diverge, it is hard to
distinguish between problems due to a bad Jacobian and problems due to
a bad initial guess. However, there is a simple clue to watch for that can
help alert the user to a bad Jacobian calculation. In most cases, Newton
converges quadratically, while “almost Newton” iterations converge linearly.
If you think you have coded Newton’s method and a plot of the residuals
shows linear behavior, look for issues with your Jacobian code!

